Newsgroup: it.comp.appl.access
Titolo: Pulsante chiusura di access

Ciao a tutti, è possibile eliminare il tasto di chisura di access (il
programma intendo non il singolo database)???
Grazie a chi mi risponderà!
Marco

 programma intendo non il singolo database)???
Sito Comune
--
Silvio
il sito comune di it.comp.appl.access:
http://www.sitocomune.com

 ha scritto nel messaggio
news:bD5G7.21964$zu5.7...@news1.tin.it...

Rimuovere il pulsante di chiusura (X), il menu e i pulsanti di
ridimensionamento di Access.
  AntoGal
Private Const GWL_STYLE = -16
Private Const WS_SYSMENU = &H80000
Private Const WS_MINIMIZEBOX = &H20000
Private Const WS_MAXIMIZEBOX = &H10000
Private Declare Function GetWindowLong Lib "user32" Alias "GetWindowLongA" _
    (ByVal hwnd As Long, ByVal nIndex As Long) As Long
Private Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" _
    (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As
Long
Private Declare Function GetParent Lib "user32" (ByVal hwnd As Long) As Long
Public Sub LockAppWindow(fForm As Form)
    ModifyAppWind fForm.hwnd
    MsgBox "Ciao ciao pulsantini!"
End Sub
Private Sub ModifyAppWind(wChildWnd)
    Dim wParentWnd%
    wParentWnd = GetParent(GetParent(wChildWnd))
    Dim PrevWinStyle As Long, NewWinStyle As Long
    PrevWinStyle = GetWindowLong(wParentWnd, GWL_STYLE)
    NewWinStyle = PrevWinStyle And Not WS_SYSMENU And _
       Not WS_MINIMIZEBOX And _
       Not WS_MAXIMIZEBOX
    PrevWinStyle = SetWindowLong(wParentWnd, GWL_STYLE, NewWinStyle)
End Sub
La funzione va richiamata così:
LockAppWindow Screen.ActiveForm

 cercato in tutte ma non lo ho trovato
Sezione General, "Rimuovere il pulsante di chiusura (X), il menu e i
pulsanti di ridimensionamento di Access." di Antogal


 ha scritto nel messaggio
news:VL7G7.22444$zu5.7...@news1.tin.it...

www.sitocomune.com/general/general032.htm
--
Andrea

 Marco

 rimpicciolire la finestra di access?
Cosa intendi x ingrandire o rimpicciolire la finestra di Access?
Prova a guardare il metodo MoveSize
DoCmd.MoveSize [destra][, giů][, larghezza][, altezza]


 Si scusa mi sono espresso male, intendevo "ingrandisci finestra" e "Riduci
a icona" ovvero i pulsanti in alto a dx.
Ciao Marco

 a icona" ovvero i pulsanti in alto a dx.
Io l'ho sempre fatto con Sendkeys, ma giustamente mi è stato fatto presente
che il funzionamento dipendeva dalla lingua del S.O., quindi ho trovato da
qualche parte questo codice che permette di minimizzare, massimizzare o
ripristinare la finestra di Access e di sapere in quale di questi stati si
trovi:
      Declare Function apiGetActiveWindow Lib "user32" Alias _
         "GetActiveWindow" () As Long
      Declare Function apiGetParent Lib "user32" Alias "GetParent" _
         (ByVal hWnd As Long) As Long
      Declare Function apiShowWindow Lib "user32" Alias "ShowWindow" _
         (ByVal hWnd As Long, ByVal nCmdShow As Long) As Long
      Global Const SW_MAXIMIZE = 3
      Global Const SW_SHOWNORMAL = 1
      Global Const SW_SHOWMINIMIZED = 2
      Declare Function apiIsIconic Lib "user32" Alias "IsIconic" (ByVal _
         hWnd As Long) As Long
      Declare Function apiIsZoomed Lib "user32" Alias "IsZoomed" (ByVal _
         hWnd As Long) As Long
      Declare Function apiMoveWindow Lib "user32" Alias "MoveWindow" _
         (ByVal hWnd As Long, ByVal X As Long, ByVal Y As Long, ByVal _
         nWidth As Long, ByVal nHeight As Long, ByVal bRepaint As Long) _
         As Long
      Function GetAccesshWnd()
         Dim hWnd As Long
         Dim hWndAccess As Long
         ' Get the handle to the currently active window.
         hWnd = apiGetActiveWindow()
         hWndAccess = hWnd
         ' Find the top window without a parent window.
         While hWnd <> 0
            hWndAccess = hWnd
            hWnd = apiGetParent(hWnd)
         Wend
         GetAccesshWnd = hWndAccess
      End Function
      Function AccessMinimize()
         AccessMinimize = apiShowWindow(GetAccesshWnd(), SW_SHOWMINIMIZED)
      End Function
      Function AccessMaximize()
         AccessMaximize = apiShowWindow(GetAccesshWnd(), SW_MAXIMIZE)
      End Function
      Function AccessRestore()
         AccessRestore = apiShowWindow(GetAccesshWnd(), SW_SHOWNORMAL)
      End Function
      Function IsAccessMaximized()
         If apiIsZoomed(GetAccesshWnd()) = 0 Then
            IsAccessMaximized = False
         Else
            IsAccessMaximized = True
         End If
      End Function
      Function IsAccessMinimized()
         If apiIsIconic(GetAccesshWnd()) = 0 Then
            IsAccessMinimized = False
         Else
            IsAccessMinimized = True
         End If
      End Function
      Function IsAccessRestored()
         If IsAccessMaximized() = False And _
               IsAccessMinimized() = False Then
            IsAccessRestored = True
         Else
            IsAccessRestored = False
         End If
      End Function
      Function AccessMoveSize(iX As Integer, iY As Integer, iWidth As _
         Integer, iHeight As Integer)
         apiMoveWindow GetAccesshWnd(), iX, iY, iWidth, iHeight, True
      End Function
In ultimo, se non mi ricordo male, Karl aveva suggerito di utilizzare
RunCommand seguito dalle costanti:
acCmdAppMaximize
acCmdAppMinimize
acCmdAppRestore
acCmdAppMove
acCmdAppSize
Ciao
